

Jobsheet IV


Filter Warna HSV

A. Tujuan

- Mahasiswa mampu menjelaskan penggunaan filter warna HSV menggunakan OpenCV untuk Processing.
- Mahasiswa mampu mengoperasikan filter warna HSV menggunakan OpenCV untuk Processing pada deteksi bola oranye.


B. Dasar Teori

Ada dua mode warna pada OpenCV, yaitu RGB (Red, Green, Blue) dan HSB (Hue, Saturation, Brightness). Berikut merupakan gambaran warna RGB dan HSB.




Gambar 4.1 Merupakan Mode Warna pada OpenCV (a) RGB (b) HSB

Pada warna RGB, setiap warna akan sulit untuk dipisahkan. Dikarenakan dari ketiga warna membutuhkan warna yang lainnya untuk memperoleh warna baru. Misalkan, ketika kita membutuhkan warna kuning, maka harus memadukan warna merah dan hijau. Warna cyan diperoleh dari penggabungan warna biru dan hijau. Sedangkan warna magenta diperoleh dari penggabungan warna biru dan merah. Namun, ketika kita membutuhkan warna putih, harus menggabungkan ketiganya. Maka, untuk menggunakan filter warna, mode warna RGB akan mempersulit proses pengambilan warna. Oleh sebab itu, pada penggunaan filter warna, digunakan mode warna HSB atau HSV.



Gambar 4.2 Filter Warna HSV pada OpenCV



Gambar 4.3 Filter Warna Oranye pada OpenCV

Pada processing penulisan penggunaan mode warna dapat dilakukan dengan kode berikut:

```
opencv.useColor( RGB );  
opencv.useColor( HSB );
```

Sedangkan untuk pengambilan setiap warna menggunakan kode berikut:

```
r = opencv.getSnapshot(opencv.getR());  
g = opencv.getSnapshot(opencv.getG());  
b = opencv.getSnapshot(opencv.getB());  
  
h = opencv.getSnapshot(opencv.getH());  
s = opencv.getSnapshot(opencv.getS());  
v = opencv.getSnapshot(opencv.getV());
```

Untuk pemfilteran grayscale dari setiap warna yang sudah diambil dapat menggunakan kode berikut tergantung warna apa yang digunakan.

```
opencv.setGray(opencv.getR().clone());  
opencv.inRange(min, max);  
  
opencv.setGray(opencv.getG().clone());  
opencv.inRange(min, max);  
  
opencv.setGray(opencv.getB().clone());  
opencv.inRange(min, max);  
  
opencv.setGray(opencv.getH().clone());  
opencv.inRange(min, max);  
  
opencv.setGray(opencv.getS().clone());  
opencv.inRange(min, max);  
  
opencv.setGray(opencv.getV().clone());  
opencv.inRange(min, max);
```

Untuk menggabungkan dua hasil filter, dapat digunakan kode berikut.

```
opencv.diff(pembeda);
```


Variable pembeda merupakan hasil filter yang akan dibandingkan dengan hasil filter sebelumnya.

C. Alat dan Bahan

- 1 set PC atau Notebook.
- Program Processing.
- Library OpenCV for Processing dan ControlP5 yang telah terunduh.
- 1 bola tenis berwarna oranye
- Karpet hijau dengan luas 4 meter persegi.


D. Langkah Kerja

1. Baca dan pahami petunjuk praktikum sebelum melakukan praktikum. Pastikan semua alat dan bahan sudah terpenuhi. Hubungi pengajar ketika terdapat masalah atau error.
2. Buat sketsa baru pada program Processing.
3. Masukkan 2 gambar dengan nama “Bola.jpg” dan “HSV.jpg” yang telah disediakan ke dalam folder sketch kita.



Gambar 4.4 Masukkan Gambar ke Folder Sketch

4. Import library ControlP5 dan OpenCV.



Gambar 4.5 Import Library ControlP5 dan OpenCV for Processing

5. Buat inisialisai berikut.

```
PImage sumber, hue, sat, val, beda, beda2;
OpenCV opencv;
Range range1, range2, range3;
ControlP5 scroll1, scroll2, scroll3, cp5;
int max1, min1, max2, min2, max3, min3;
int sliderValue = 10;
```

6. Buat setting project sesuai dengan kode berikut.

```
void setup() {
  sumber = loadImage("HSV.jpg");

  opencv = new OpenCV(this, sumber);
  size(640+320, 480+80);
  background(#005250);

  opencv.useColor(HSB);
  scroll1 = new ControlP5(this);
  scroll2 = new ControlP5(this);
  scroll3 = new ControlP5(this);
  cp5 = new ControlP5(this);

  cp5.addSlider("sliderValue")
    .setPosition(640, 485)
    .setRange(0, 255)
    ;

  range1 = scroll1.addRange("Min H, Range H, Max H")
    .setBroadcast(false)      .setPosition(640/4, 485)
    .setSize(320, 20)         .setHandleSize(10)
    .setRange(0, 255)         .setRangeValues(50, 100)
    .setBroadcast(true)      ;

  range2 = scroll2.addRange("Min S, Range S, Max S")
    .setBroadcast(false)      .setPosition(640/4, 510)
    .setSize(320, 20)         .setHandleSize(10)
    .setRange(0, 255)         .setRangeValues(50, 100)
    .setBroadcast(true)      ;

  range3 = scroll3.addRange("Min V, Range V, Max V")
    .setBroadcast(false)      .setPosition(640/4, 535)
    .setSize(320, 20)         .setHandleSize(10)
    .setRange(0, 255)         .setRangeValues(50, 100)
    .setBroadcast(true)      ;

  println("-----");
  println("|           Filter Warna HSB OpenCV           |");
  println("|  Praktikum Pemrosesan Sinyal @G4.206  |");
  println("|           Universitas Negeri Malang           |");
  println("-----");
}
```

7. Masukkan kode berikut pada void draw()

```
void draw() {
    opencv.loadImage(sumber);
    opencv.useColor(HSB);

    opencv.setGray(opencv.getH().clone());
    opencv.inRange(min1, max1);
    opencv.blur(12);
    opencv.threshold(50);
    opencv.dilate();
    opencv.erode();
    hue=opencv.getSnapshot();

    opencv.setGray(opencv.getS().clone());
    opencv.inRange(min2, max2);
    opencv.blur(12);
    opencv.threshold(50);
    opencv.dilate();
    opencv.erode();
    sat=opencv.getSnapshot();

    opencv.diff(hue);
    opencv.threshold(sliderValue);
    opencv.invert();
    beda=opencv.getSnapshot();

    opencv.setGray(opencv.getV().clone());
    opencv.inRange(min3, max3);
    opencv.blur(12);
    opencv.threshold(50);
    opencv.dilate();
    opencv.erode();
    val=opencv.getSnapshot();

    opencv.diff(beda);
    opencv.threshold(sliderValue);
    opencv.invert();
    beda2=opencv.getSnapshot();

    image(sumber, 0, 0, 640/2, 480/2);
    image(hue, 640/2, 0, 640/2, 480/2);
    image(sat, 640, 0, 640/2, 480/2);
    image(val, 0, 480/2, 640/2, 480/2);
    image(beda, 640/2, 480/2, 640/2, 480/2);
    image(beda2, 640, 480/2, 640/2, 480/2);

    fill(255, 0, 0);
    text("Sumber", 640/2 - 100, 20 );
    text("Hue", 640 - 100, 20 );
    text("Saturation", 960 - 100, 20 );
    text("Value", 640/2 - 100, 480/2 + 20 );
    text("Gabungan H & S", 640 - 150, 480/2 + 20 );
    text("Gabungan H,S & V", 960 - 150, 480/2 + 20 );
    text("Nama: Yogi Dwi Mahandi", 640 - 150, 480/2 + 40 );
}
```

8. Tambahkan fungsi controEvent pada sketch kita. Letakkan setelah dan di luar void draw().

```

void controlEvent(ControlEvent theControlEvent) {
  if (theControlEvent.isFrom("Min H, Range H, Max H")) {
    min1 = int(theControlEvent.getController().getArrayValue(0));
    max1 = int(theControlEvent.getController().getArrayValue(1));
  }

  if (theControlEvent.isFrom("Min S, Range S, Max S")) {
    min2 = int(theControlEvent.getController().getArrayValue(0));
    max2 = int(theControlEvent.getController().getArrayValue(1));
  }

  if (theControlEvent.isFrom("Min V, Range V, Max V")) {
    min3 = int(theControlEvent.getController().getArrayValue(0));
    max3 = int(theControlEvent.getController().getArrayValue(1));
  }
}
    
```

9. Run sketch
10. Ubah nilai min, range, dan max sesuai dengan Tabel 4.1. Amati apa yang terjadi. Beri screenshot pada tiap filter gabungannya.
11. Ubah sumber gambar menggunakan “Bola.jpg”.
12. Ubah nilai filter HSV sehingga hanya bola oranye yang dapat terdeteksi.
13. Masukkan pada Tabel 4.2.
14. Modifikasi kode pada sketch sehingga sumber berasal dari webcam laptop Anda.
15. Lakukan filter bola oranye pada karpet hijau yang telah disediakan secara bergantian tiap kelompok.
16. Ubah posisi pengindraan bola. Amati dan masukkan pada tabel.

E. Hasil Pengamatan

Tabel 4.1 Hasil Pengamatan Min, Range, Max Bola

No	Hue		Saturation		Value		Filter H + S	Fiter H + S + V
	Min	Max	Min	Max	Min	Max		
1	34	54	80	105	110	140		
2	85	100	160	185	30	60		
3	150	160	95	120	218	248		
4	170	180	70	95	225	255		

Tabel 4.2 Hasil Pengamatan Filter HSV pada Bola

Screenshot Hasil Filter	Hue		Saturation		Value	
	Min	Max	Min	Max	Min	Max
Bola.jpg						
Webcam						

F. Analisa Data

1. Berdasarkan pada Tabel 4.1, jelaskan secara singkat fungsi penggabungan filter H + S !
2. Berdasarkan pada Tabel 4.1, jelaskan secara singkat fungsi penggabungan filter H + S + V !
3. Mengapa pada pemfilteran bola oranye, harga H, S dan V selalu berubah pada tiap posisi pengindraan camera?

G. Tugas

1. Modifikasi program diatas sesuai dengan kreatifitas Anda, lampirkan melalui Youtube dan kirim tautannya ke yogidwimahandi@gmail.com

H. Kesimpulan

.....
.....
.....